Haupt und Unterformular So klappt der Zugriff per VBA
Versionen: Access 2007, 2003, 2002/XP, 2000 und 97

Beim Aufbau Ihrer Datenbanken stehen Sie immer wieder vor dem Problem, vom Hauptformular aus einen Wert in einem Feld eines Unterformulars auslesen oder setzen zu müssen. Genauso häufig ist es notwendig, nach Änderungen im Unterformular einzelne Felder im Hauptformular zu ändern.

Felder in einem Unterformular erreichen Sie am einfachsten über die "Forms"- und die "Controls"-Auflistung: In "Forms" verwaltet Access alle geöffneten Formulare, so dass hier auch der Ausgangspunkt für den Zugriff auf ein beliebiges geöffnetes Formular liegt. Wichtig dabei: In "Forms" verwaltet Access nur Hauptformulare! Obwohl ein in einem Hauptformular genutztes Unterformular auch ein geöffnetes Formular darstellt, ist der direkte Zugriff über "Forms" nicht möglich.

In der "Controls"-Auflistung verwaltet jedes Formular seine Steuerelemente. Sowohl das Haupt- als auch das Unterformular verfügen über eine solche Auflistung. Um ein Unterformular in einem Hauptformular zu adressieren, verwenden Sie die folgende Syntax:

Forms("HFoName").Controls("UFoName")

"HFoName" ist dabei der Name des Hauptformulars und "UFoName" der Name des Unterformular-Steuerelementes. Sie greifen also nicht über den Namen des Unterformulars so wie er im Datenbankfenster angezeigt wird, sondern über den Namen des Steuerelementes, mit dem das Unterformular in das Hauptformular eingebunden ist, auf das Unterformular zu. Oftmals sind diese Namen zwar identisch, aber wenn eine Lösung einmal nicht wie erwartet funktioniert, prüfen Sie als erstes, ob der Namen des Unterformulars eventuell vom Namen des Unterformular-Steuerelementes abweicht.

Der Zugriff auf ein Feld in einem Unterformular wird am besten an einem Beispiel deutlich: Öffnen Sie einmal das Formular "Bestellungen" in der Beispieldatenbank "Nordwind". Wechseln Sie dann mit Alt+F11 in die VBA-Entwicklungsumgebung oder öffnen Sie ein beliebiges Modul, lassen Sie mit Strg+G den Direktbereich anzeigen und geben Sie dort folgende Anweisung ein:

? Forms("Bestellungen").Controls("Bestellungen 
Unterformular").Controls("Anzahl")

Access gibt daraufhin die Bestellmenge des ersten Datensatzes gemäß Feld "Anzahl" im Direktbereich aus.

Aufbauend auf diesen Grundlagen kann nun zum Beispiel eine Schaltfläche im Hauptformular angelegt werden, die eine neue Bestellung mit einem "Angebot der Woche" vorbelegt:

Private Sub btnAdW_Click()

  With Forms("Bestellungen"). _
  Controls("Bestellungen Unterformular")
    .Controls("Artikel-Nr") = 30
    .Controls("Rabatt") = "25%"
    .Controls("Einzelpreis") = 12.5
    .Controls("Anzahl").SetFocus
    .SetFocus
  End With

End Sub

Diese Anweisungen tragen die Artikelnummer "30", einen Rabatt von 25% und einen Einzelpreis von 12,50 Euro ein und setzen dann den Focus auf das Feld "Anzahl" sowie auf das Unterformular-Steuerelement selbst, damit die Bestellmenge direkt eingegeben werden kann.

Ähnlich einfach können Sie aus dem Unterformular heraus Eingaben oder Einstellungen im Hauptformular ändern. Beispiel: Ab einer bestimmten Bestellmenge soll automatisch der Versand per "United Package" aktiviert werden. Dazu wird der im Feld "Anzahl" eingegeben Wert über die Ereignisprozedur "Nach Aktualisierung" geprüft und das Hauptformular ggf. entsprechend angepasst:

Private Sub Anzahl_AfterUpdate()
  Dim lngAnz As Long

  On Error Resume Next
  lngAnz = CLng(Me.Anzahl)
  If lngAnz >= 100 Then
    Forms("Bestellungen").Controls("VersandÜber") = 2
  End If

End Sub

Diese Anweisungen lesen den aktuellen Wert aus dem Feld "Anzahl" aus und konvertieren ihn in eine Long Integer-Zahl. Die folgende IF-Abfrage prüft dann, ob dieser Wert größer oder gleich "100" ist. Ist das der Fall, wird im Hauptformular der Inhalt der Optionsgruppe "Versand per" (Steuerelemente= "VersandÜber") auf den Wert "2" gesetzt, der "United Package" entspricht.

Damit sind zunächst einmal die Grundlagen vermittelt, wie auf Felder eines Unterformulars vom Hauptformular aus und umgekehrt zugegriffen werden kann. In einer der nächsten Ausgaben von Access Weekly stellen wir Ihnen eine Technik vor, mit der Sie aus dem Hauptformular Funktionen in einem Unterformular aufrufen können (und umgekehrt). Damit lassen sich dann komplexere Lösungen wie zum Beispiel eine Neupositionierung im Hauptformular abhängig von Eingaben oder Einstellungen im Unterformular realisieren.

